


POSTERS

E-Mail z/OS® SAS®/MXG® Tape Management Reports with ODS Traffic Lighting

Neal Musitano Jr.
Department of Veterans Affairs
Information Technology Center
Philadelphia, Pennsylvania

ABSTRACT



This paper displays an e-mailed tape management report with “Traffic Lighting” using the BASE SAS® Output Delivery System (ODS). The report is created and e-mailed via a z/OS® batch job using the BASE SAS® JCL procedure. E-mail reports with ODS traffic lighting can be generated from your choice of input data. The input data for this example is from the daily SAS/MXG® Performance Database PDB.TAPES,

The tape management e-mail report with ODS Traffic Lighting in red, yellow and green colors is generated as an HTML file using the BASE SAS® Output Delivery System (ODS) and PROC PRINT. Reports can also be generated via other procedures such as PROC REPORT.

INTRODUCTION

The IBM System Management Facility (SMF) record type 21 is written when a magnetic tape is dismounted. This SMF record contains tape I/O and tape error information that is processed into the MXG® PDB.TAPES dataset. The advantage of using MXG® software is that it converts raw performance data from IBM and many other vendors into SAS® readable format for processing by SAS® jobs.

SAS/MXG software processes the z/OS SMF/RMF performance data into daily, weekly and monthly MXG® Computer Performance Databases. These databases contain numerous SAS® datasets. Each SAS® dataset contains performance variables in a SAS® readable format. MXG® also provides numerous SAS® examples for analysis and/or plots, from utilizing BASE SAS® to using SAS/GRAPH®.

PREPARATION

Operations managers wanted a tape management report identifying tapes with read/write errors. Tape drives are cleaned regularly and reports are generated to show error rates for tape drives. However, this report identifies specific tape volser's and uses ODS “Traffic Lights” to highlight the tape error counts. The setting for the green, yellow, red traffic light values can be a topic for continuing additional discussions.

This example used PROC PRINT to generate the report. To use ODS ‘Traffic Lighting’ with PROC PRINT the red, yellow, green, etc. traffic light background colors needed to be specified using PROC FORMAT.

Additionally two SAS e-mail options were used in the Job Control Language JCL for the batch job. The SAS **OPTIONS=‘EMAILSYS=SMTP EMAILHOST=123.456.789.2’** specifies the SMTP e-mail interface and the TCP/IP address of the mainframe e-mail server. The option ‘EMAILPORT=25’ to identify the default e-mail port number, might be needed at some sites. However, depending on your base SAS procedure config parameter options and system TCP/IP installation you may want to check with your computer site if specifying the parameters in the JCL is necessary or to verify your computer site's specific parameter information.

POSTERS

TRAFFIC LIGHT E-MAIL PROGRAM DETAILS

The first part of SAS code used to produce the report is displayed in Figure #1.

PROC FORMAT is used to define the red, yellow and green traffic lights. Two traffic light formats are defined in this example. Traffic light MY_LITEA is a red, yellow and green light that will be used for temp read/clean errors. Traffic light MY_LITEB a red/green light will be used to highlight permanent errors.

The traffic lights are implemented in PROC PRINT as a background style on the VAR statement. Setting the values for the lights can be a topic of continuing discussion. Additionally even a perm error can be a problem with the tape drive and not the tape.

PROC TEMPLATE is used to edit the default style to change the overall background color of the HTML report from the default gray background to a light blue color CXCCFFFF. The style MY_STYLE is used in the ODS statement.


Figure #1 Define The Traffic Lights

```
/* TAPE MANAGEMENT REPORT - ODS TRAFFIC LIGHTS USED */
/* REPORT SENT VIA E-MAIL */

PROC FORMAT;
VALUE MY_LITEA /* TRAFFIC LIGHT "MY_LITEA" FOR TEMP ERRORS */
  LOW-1 = LIGHTGREEN /* USED IN PROC PRINT VAR STATEMENT AS A */
  2-6 = YELLOW /* BACKGROUND SYTLE OPTION */
  7-HIGH = RED;

VALUE MY_LITEB /* TRAFFIC LIGHT "MY_LITEB" FOR PERM ERRORS */
  0 = LIGHTGREEN /* USED IN PROC PRINT */
  OTHER = RED; /* FOR "PERM" RED GREEN LIGHTS */

/* CHANGE THE REPORT BACKGROUND COLOR TO LIGHT BLUE COLOR IN FIG 2 */
/* THE STYLE MY_STYLE IS USED ON THE ODS STATEMENT - IN FIGURE 3A */
PROC TEMPLATE;
EDIT STYLES.DEFAULT AS STYLES.MY_STYLE;
  STYLE COLOR_LIST /
    'bgA' = CXCCFFFF;
  END;
RUN;
```




The "Traffic Light" HTML e-mail display on the next page is being viewed from Microsoft Outlook. Note that some internet service providers, have firewalls that prevent HTML e-mails or have their own e-mail set up that might perform unexpected edits on HTML e-mail.

The tape management report displays information about several types of tape errors. When the error count increases the traffic light will highlight the error with the selected color. Tapes with high error rates are often candidates for replacement when scratch dates come up. Additional reports are also used to identify tape drives with a high percentage of error rates, indicating the tape drive(s) that may need cleaning or service.

POSTERS

Figure #2 z/OS BASE SAS® 'Traffic Light' Tape Management Report



ODS “TRAFFIC LIGHTS” HIGHLIGHT THE E-MAIL REPORT ON TAPE ERRORS

Figure #2 displays a print screen of the tape management e-mail report generated from BASE/SAS® using the Output Delivery System (ODS). A modified test MXG® PDB was used for input.

The write erase actions and the read cleaner actions, are highlighted in red, yellow and green by using the MY_LITEA format as a background style option on the VAR statement in PROC PRINT. The permanent read errors and the permanent write errors are highlighted in red and green using the MY_LITEB format.

The variable labels instead of the variable names are displayed in the report heading by using the SPLIT option in PROC PRINT.

POSTERS

The SAS code for the report continues in FIGURE 3A-B.

The DATA step creates a “MY_TAPE_MANAGEMENT_DATA ” SAS dataset. The SET statement was used to read the daily MXG® PDB.TAPES dataset that was used for input to the program. The PDB.TAPES dataset contains basic tape information on tape I/O counts, error counts, tape device, timestamp and more.

PROC SORT is used to sort the work dataset MY_TAPE_MANAGEMENT_DATA. For this report, operations tape managers wanted the report by “DEVICE” groups then descending by tape error counts.

The FILENAME statement defines “MY_FILE” will be written as an EMAIL file. Additional options defining “FROM”, “TO”, “CC”, “REPLYTO”, and “SUBJECT” are specified and the file content is defined as “TEXT/HTML” file type. Note, that multiple e-mail id’s can be specified in the “TO” and “CC” fields.

The Output Delivery System (ODS) is creating a HTML file. Note the body “MY_FILE” is referenced in the FILENAME statement as an EMAIL file. The STYLE of MY_STYLE is from the PROC TEMPLATE edit of the default style.

Figure #3A Read the SAS/MXG® TAPES dataset and define the EMAIL file options

```
/* INPUT THE TAPE INFORMATION - ONLY REPORT ON TAPES WITH ERRORS */


DATA MY_TAPE_MANAGEMENT_DATA;
  SET PDB.TAPES;
  IF VOLSER = '000000000000'X THEN DELETE; /* DELETE IF NO TAPE VOLSER */
  IF ERRORS GE 1 THEN OUTPUT; /* ONLY TAPES WITH ERRORS */
RUN;

/* SORT THE TAPE INFORMATION BY DEVICE AND DESCENDING ERROR COUNT */
PROC SORT DATA= MY_TAPE_MANAGEMENT_DATA;
  BY DEVICE DESCENDING ERRORS;
RUN;

/* THE REPORT WILL BE E-MAILED */
/* DEFINE THE E-MAIL FILE AND E-MAIL DETAILS */

FILENAME MY_FILE EMAIL
  FROM= ("SASMXG-TAPEMONITOR")
  TO= ("PERSONAEMAILID@VA.GOV"
      "PERSONBEMAILID@VA.GOV")
  CC= (
      "YOURMAILID@VBA.VA.GOV")
  REPLYTO= ("YOUREMAILID@ABCD.COM")
  SUBJECT="SAS MXG Tape Error VOLSER Report &SYSDATE9"
  TYPE="TEXT/HTML" ;
RUN;

/* ODS STATEMENT DEFINING THE BODY AND USING MY_STYLE FROM PROC TEMPLATE */
ODS HTML BODY=MY_FILE STYLE=MY_STYLE;
```



POSTERS


PROC PRINT is used to generate the report, since it is easy to use for ODS “Traffic Lighting” and can be expanded on for future report modifications. Input to PROC PRINT is the sorted “MY_TAPE_MANAGEMENT_DATA ” file from the previous PROC SORT.

The traffic lights are implemented as a background color with the style option on the VAR statements. This permitted using different “Traffic Lighting” for the different variables. MY_LITEA is used for “CLEAN” and “ERASE” errors while MY_LITEB is used for the “PERMREAD” and “PERMWRT” errors.

Two title statements TITLE1 and TITLE2 are used with blue text on a green background. The FOOTNOTE1 statement is used to place a footnote at the bottom left of the report. The SAS® jobid and day, date and time that the report job started execution are displayed in the footnote as blue text.

Figure #3B Using PROC PRINT to produce the 'traffic lighted' tape report

```
/* USE PROC PRINT TO PRODUCE THE TAPE TRAFFIC LIGHT REPORT */  
  
PROC PRINT DATA=MY_TAPE_MANAGEMENT_DATA SPLIT='*'   
  STYLE={BACKGROUND=BLUE}; /* BLUE LINES AROUND CELLS */  
  BY DEVICE;  
  
  ID DEVICE ;  
  VAR DEVNR VOLSER / STYLE={FONT_WEIGHT=BOLD};  
  VAR ERASE CLEAN / STYLE={FONT_WEIGHT=BOLD BACKGROUND=MY_LITEA.};  
  VAR PERMREAD PERMWRT / STYLE={FONT_WEIGHT=BOLD BACKGROUND=MY_LITEB.};  
  TITLE1 COLOR=BLUE BCOLOR=CX00FF00 h=2.0  
    'The following Tapes - Have at least one error';  
  TITLE2 COLOR=BLUE BCOLOR=CX00FF00 h=2.0  
    ' and may be candidates for replacement';  
  
  FOOTNOTE1 COLOR=BLUE JUSTIFY=LEFT HEIGHT=1.8  
    " Report from job = &sysjobid on &Sysday &sysdate9 at &systemtime";  
  
  /* NOTE USING DOUBLE QUOTES IN FOOTNOTE */  
  
RUN;  
ODS HTML CLOSE;
```



Note: A modified test MXG/PDB was used for the tape information displayed. The device types, device numbers and tape volumes were changed from actual values for security reasons – not to include the actual values of detailed computer performance data in the paper. No actual values of production computer performance information were used in the above report.

POSTERS

CONCLUSION

The z/OS Base SAS JCL procedure is easily used to generate and e-mail tape management reports with ODS traffic lighting. Using PROC FORMAT to define the “traffic lights” and PROC PRINT to implement them permitted the program development effort to be accomplished in a very timely manner.

Producing an ODS “Traffic Light” tape management report from BASE SAS® using PROC PRINT provided operations tape managers with what they requested. The reports are e-mailed to specific individuals on time, eliminating lost and misrouted printed reports, and improved the ease of use for managers and technicians to review computer performance and tape activity.

REFERENCES

IBM z/OS - MVS System Management Facilities (SMF) – SA22-7630-06
Merrill's Expanded Guide to Computer Performance Evaluation Using the SAS System
Merrill's Expanded Guide Supplement
Technical Newsletters for Users of MXG®
MXG Archives at WWW.MXG.COM
SUGI Paper Presentation Guidelines at <http://support.sas.com/usergroups/sugi/sugi30/package/guidelines.pdf>
SAS Technical Support at <http://support.sas.com/>
SAS-L archives at <http://www.listserv.uga.edu/>
SAS z/OS Examples and Tips at <http://support.sas.com/documentation/hosts/mainframe>
SAS ONLINE Documentation at <http://support.sas.com/documentation/onlinedoc>
SUGI ONLINE Proceeding at <http://support.sas.com/usergroups/sugi/proceedings/index.html>

CONTACT INFORMATION

Your comments and questions are welcome.

Neal Musitano Jr.
Department of Veterans Affairs
Information Technology Center-284/31
P. O. BOX 7545
Philadelphia, PA. 19101-7545

Phone 215-842-2000 ext. 4102
FAX 215-381-3456
EMAIL Neal.Musitano@vba.va.gov

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.